```
Imports System.Net
Imports System.Net.Sockets
Imports System.Text

Public Class Form1

    Public Const DEFAULT_HOST As String = "localhost"
    Public Const DEFAULT_PORT As Short = 1831
    Public Const DEFAULT_SHOW_SOCK_MSGS As Boolean = False

    Public Enum RequestType As Integer
        DateTime = 1
        RandomCookie = 2
        SpecificCookie = 3
    End Enum

    Public Const DEFAULT_REQUEST_TYPE As RequestType = RequestType.RandomCookie
    Public Const DEFAULT_COOKIE_NUM As Integer = 1

    Private UserHost As String = DEFAULT_HOST
    Private UserPort As Short = DEFAULT_PORT
    Private UserShowSockMsgs As Boolean = DEFAULT_SHOW_SOCK_MSGS
    Private UserRequest As RequestType = DEFAULT_REQUEST_TYPE
    Private UserCookieNum As Integer = DEFAULT_COOKIE_NUM

    Private Sub ButtonGo_Click(ByVal sender As System.Object, ByVal e As System.EventArgs) Handles ButtonGo.
    Click

        UserHost = TextBoxHost.Text

        UserPort = CShort(NumericUpDownPort.Value)

        UserShowSockMsgs = CheckBoxSockMsgs.Checked

        If RadioButtonDateTime.Checked Then
            UserRequest = RequestType.DateTime
        ElseIf RadioButtonRandomCookie.Checked Then
            UserRequest = RequestType.RandomCookie
        Else
            UserRequest = RequestType.SpecificCookie
            UserCookieNum = NumericUpDownCookie.Value
        End If

        DoWinSock()

    End Sub

    Private Sub DoWinSock()

        Try

            ' Data buffer for incoming data.
            Dim bytes(1024) As Byte

            ' Connect to a remote device.

            ' Establish the remote endpoint for the socket.
            Dim ipHostInfo As IPHostEntry = Dns.GetHostEntry(UserHost)
            Dim ipAddress As IPAddress = ipHostInfo.AddressList(0)
            Dim remoteEP As New IPEndPoint(ipAddress, UserPort)

            ' Create a TCP/IP socket.
            Dim sender As New Socket(AddressFamily.InterNetwork, _
                SocketType.Stream, ProtocolType.Tcp)

            ' Connect the socket to the remote endpoint.
            sender.Connect(remoteEP)
```

```
            ShowSocketMsg("Socket connected to " & sender.RemoteEndPoint.ToString())

            ' Encode the data string into a byte array.
            Dim req As Integer = UserRequest

            Dim msg As Byte() = _
                Encoding.ASCII.GetBytes(req.ToString() & UserCookieNum.ToString() & "<EOF>")

            ' Send the data through the socket.
            Dim bytesSent As Integer = sender.Send(msg)

            ' Receive the response from the remote device.
            Dim bytesRec As Integer = sender.Receive(bytes)

            Dim sCookie As String = _
                Encoding.ASCII.GetString(bytes, 0, bytesRec)

            If UserRequest = RequestType.DateTime Then

                ShowDateAndTime(sCookie)
            Else

                Dim sb As StringBuilder = New StringBuilder
                Dim ch As Char

                For c As Integer = 0 To sCookie.Length() - 1

                    ch = sCookie.Chars(c)

                    If ch = "|" Then
                        sb.Append(vbCrLf)
                    Else
                        sb.Append(ch)
                    End If
                Next

                If UserRequest = RequestType.RandomCookie Then

                    ShowRandomCookie(sb.ToString())

                Else

                    showSpecificCookie(sb.ToString())

                End If


            End If

            ' Release the socket.
            sender.Shutdown(SocketShutdown.Both)
            sender.Close()

        Catch ex As Exception

            ShowSocketError(ex.ToString())

        End Try

    End Sub

    Private Overloads Sub Output(ByVal message As String, ByVal colour As Color, _
        ByVal bold As Boolean)

        With RichTextBoxMsgs

            .SelectionStart = .Text.Length
```

```
            .SelectionColor = colour

            If bold Then
                .SelectionFont = New Font(.SelectionFont.FontFamily, .SelectionFont.Size, _
                                          FontStyle.Bold)
            Else
                .SelectionFont = New Font(.SelectionFont.FontFamily, .SelectionFont.Size, _
                                          FontStyle.Regular)
            End If

            .AppendText(message)
            .ScrollToCaret()

        End With

    End Sub

    Private Overloads Sub Output(ByVal message As String, ByVal colour As Color)

        Output(message, colour, False)

    End Sub

    Private Overloads Sub Output(ByVal message As String)

        Dim defColour As Color = RichTextBoxMsgs.ForeColor

        Output(message, defColour, False)

    End Sub

    Private Sub EndOutput()

        With RichTextBoxMsgs

            .AppendText(System.Environment.NewLine)
            .ScrollToCaret()

        End With

    End Sub

    Private Sub ShowSocketMsg(ByVal sMsg As String)

        If UserShowSockMsgs Then

            Output(sMsg, Color.Purple)
            EndOutput()

        End If

    End Sub

    Private Sub ShowSocketError(ByVal sErr As String)

        Output(sErr, Color.Red, True)
        EndOutput()

    End Sub

    Private Sub ShowDateAndTime(ByVal sDateTime As String)

        Output("Server date & time is ", Color.Purple)
        Output(sDateTime, Color.Orange, True)
        EndOutput()

    End Sub
```

```
    Private Sub ShowRandomCookie(ByVal sCookie As String)

        Output("Random cookie:", Color.Purple)
        EndOutput()
        Output(sCookie, Color.Blue)
        EndOutput()

    End Sub

    Private Sub showSpecificCookie(ByVal sCookie As String)

        Output("Specific cookie #", Color.Purple)
        Output(UserCookieNum.ToString(), Color.DarkBlue, True)
        Output(":", Color.Purple)
        EndOutput()
        Output(sCookie, Color.Blue)
        EndOutput()

    End Sub

    Private Sub Form1_Load(ByVal sender As System.Object, ByVal e As System.EventArgs) Handles MyBase.Load

        TextBoxHost.Text = DEFAULT_HOST
        NumericUpDownPort.Value = DEFAULT_PORT

        If DEFAULT_REQUEST_TYPE = RequestType.DateTime Then
            RadioButtonDateTime.Checked = True
        ElseIf DEFAULT_REQUEST_TYPE = RequestType.RandomCookie Then
            RadioButtonRandomCookie.Checked = True
        Else
            RadioButtonSpecificCookie.Checked = True
        End If
        NumericUpDownCookie.Value = DEFAULT_COOKIE_NUM

    End Sub

    Private Sub ButtonClear_Click(ByVal sender As System.Object, ByVal e As System.EventArgs) Handles
    ButtonClear.Click

        RichTextBoxMsgs.Clear()

    End Sub
End Class
```